



ISPC

Mesurer les inégalités en santé

Trousse d'outils

Macros SAS et notes méthodologiques



Institut canadien
d'information sur la santé
Canadian Institute
for Health Information

La production du présent document est rendue possible grâce à un apport financier de Santé Canada et des gouvernements provinciaux et territoriaux. Les opinions exprimées dans ce rapport ne représentent pas nécessairement celles de Santé Canada ou celles des gouvernements provinciaux et territoriaux.

Tous droits réservés.

Le contenu de cette publication peut être reproduit tel quel, en tout ou en partie et par quelque moyen que ce soit, uniquement à des fins non commerciales pourvu que l'Institut canadien d'information sur la santé soit clairement identifié comme le titulaire du droit d'auteur. Toute reproduction ou utilisation de cette publication et de son contenu à des fins commerciales requiert l'autorisation écrite préalable de l'Institut canadien d'information sur la santé. La reproduction ou l'utilisation de cette publication ou de son contenu qui sous-entend le consentement de l'Institut canadien d'information sur la santé, ou toute affiliation avec celui-ci, est interdite.

Pour obtenir une autorisation ou des renseignements, veuillez contacter l'ICIS :

Institut canadien d'information sur la santé

495, chemin Richmond, bureau 600

Ottawa (Ontario) K2A 4H6

Téléphone : 613-241-7860

Télécopieur : 613-241-8120

www.icis.ca

droitdauteur@icis.ca

© 2018 Institut canadien d'information sur la santé

This publication is also available in English under the title *Measuring Health Inequalities: A Toolkit — SAS Macros and Methodology Notes*.

Table des matières

Macros SAS	4
1. %Calculate_Stratified_Rates	4
2. %Calculate_Inequality_Measures	4
Notes méthodologiques	5
1. Taux normalisés selon l'âge	5
2. Mesures simples de l'inégalité	5
3. Mesures complexes de l'inégalité	7
Points à considérer lors de la déclaration des mesures complexes de l'inégalité	8
Annexe 1 : Macro SAS — %Calculate_Stratified_Rates	9
Annexe 2 : Macro SAS — %Calculate_Inequality_Measures	20
Référence	27

Le présent guide donne un aperçu des 2 macros SAS que vous pouvez utiliser pour mesurer les inégalités en santé. Il comprend aussi des notes méthodologiques, des formules et des points à prendre en considération dans la déclaration des résultats.

Macros SAS

1. %Calculate_Stratified_Rates

Cette macro calcule les taux bruts et normalisés selon l'âge, stratifiés par quintile de revenu ou par emplacement géographique (région urbaine ou rurale/éloignée). Les taux stratifiés sont aussi calculés en fonction du sexe (hommes, femmes et les 2 sexes) et du niveau de déclaration (à l'échelle du pays ou des provinces et territoires). Ce code peut être modifié pour calculer les taux stratifiés en fonction d'autres facteurs de stratification de l'équité.

Pour utiliser la macro

- Créez les 3 fichiers d'entrée suivants, formatés selon la convention décrite dans le document Excel [Mesurer les inégalités en santé : trousse d'outils — formats des fichiers d'entrée des macros SAS](#) :
 1. Cas liés à l'indicateur;
 2. Dénominateurs de population;
 3. Population de référence stratifiée selon vos groupes d'âge (p. ex. groupes d'âge de 5 ans).
- Définissez les paramètres de la macro, y compris la tranche d'âge pour l'indicateur.
- Lancez la macro %Calculate_Stratified_Rates (annexe 1).
- Vérifiez que les données générées sont complètes et que le journal est exempt d'erreurs et d'avertissements.

2. %Calculate_Inequality_Measures

Cette macro calcule les mesures des inégalités, dont le ratio des taux, la différence des taux, la réduction potentielle des taux (RPT) et le nombre de personnes à risque au sein de la population (NPRP), pour des indicateurs de santé stratifiés par quintile de revenu ou par emplacement géographique (région urbaine ou rurale/éloignée). Les résultats sont générés en fonction du sexe (hommes, femmes et les 2 sexes) et du niveau de déclaration (à l'échelle du pays ou des provinces et territoires).

Pour utiliser la macro

- Créez un fichier d'entrée formaté selon la convention décrite dans le document Excel [Mesurer les inégalités en santé : trousse d'outils — formats des fichiers d'entrée des macros SAS](#). Notez que les données générées après l'exécution réussie de la macro %Calculate_Stratified_Rates seront correctement formatées.
- Lancez la macro %Calculate_Inequality_Measures (annexe 2).
- Vérifiez que les données générées sont complètes et que le journal est exempt d'erreurs et d'avertissements.

Notes méthodologiques

1. Taux normalisés selon l'âge

Les taux des indicateurs ont été normalisés selon l'âge au moyen d'une méthode de normalisation directe qui utilise la population canadienne de 2011 (selon le Recensement de 2011) comme population de référence. La normalisation est effectuée en fonction de groupes d'âge de 5 ans.

Le **taux normalisé selon l'âge** est calculé à l'aide de la formule suivante :

$$\text{Taux normalisé selon l'âge} = \frac{\text{numérateur}}{\text{dénominateur}} \times \text{pondération} \times \text{multiplicateur}$$

où la pondération est celle de la population de référence et le multiplicateur correspond à la valeur pour laquelle vous souhaitez présenter les taux (p. ex. par 100 000 habitants).

La **variance** est calculée à l'aide de la formule suivante :

$$\text{Variance (taux)} = \text{pondération}^2 \times \text{taux}_{\text{brut}} \times \frac{(100\,000 - (\text{taux}_{\text{brut}}))}{\text{population}}$$

2. Mesures simples de l'inégalité

Le **ratio des taux** est une mesure de l'inégalité *relative* entre sous-groupes. On le calcule en divisant le taux d'un sous-groupe (habituellement celui affichant le taux le moins souhaitable) par le taux d'un autre sous-groupe (habituellement celui affichant le taux le plus souhaitable).

Exemple : Mesurer les inégalités liées au revenu, où Q1 est le quintile de revenu le plus faible et Q5, le quintile de revenu le plus élevé.

$$Q1 \div Q5 = 10 \div 5 = 2,00$$

Interprétation : Le taux de l'affection X est 2 fois plus élevé chez les Canadiens du quintile de revenu le plus faible que chez ceux du quintile de revenu le plus élevé.

La **variance** est calculée à l'aide de la formule suivante :

$$\text{Variance} \left(\log \left(\frac{\text{taux}_{Q1}}{\text{taux}_{Q5}} \right) \right) = \frac{\text{variance} (\text{taux}_{Q1})}{\text{taux}_{Q1}^2} + \frac{\text{variance} (\text{taux}_{Q5})}{\text{taux}_{Q5}^2}$$

L'**intervalle de confiance de 95 % du ratio des taux** est calculé à l'aide de la formule $e^{(\log (\text{ratio des taux}) \pm 1,96\sqrt{\text{variance} (\text{ratio des taux})})}$.

Le ratio des taux est considéré comme significativement différent de l'hypothèse nulle (ratio des taux = 1) du point de vue statistique si l'intervalle de confiance de 95 % ne comprend pas la valeur 1.

La **différence des taux** est une mesure de l'inégalité *absolue* entre sous-groupes. On la calcule en soustrayant le taux d'un sous-groupe du taux d'un autre sous-groupe (habituellement le taux le plus élevé du taux le plus faible).

Exemple : $Q1 - Q5 = 10 - 5 = 5$

Interprétation : L'affection X atteint 5 Canadiens de plus dans le quintile de revenu le plus faible que dans le quintile de revenu le plus élevé.

La **variance** est calculée à l'aide de la formule suivante :

$$\text{Variance} (\text{différence des taux}) = \text{variance} (\text{taux}_{Q1}) + \text{variance} (\text{taux}_{Q5})$$

L'**intervalle de confiance de 95 % de la différence des taux** est calculé à l'aide de la formule $\text{différence des taux} \pm 1,96\sqrt{\text{variance} (\text{différence des taux})}$.

La différence des taux est considérée comme significativement différente de l'hypothèse nulle (différence des taux = 0) du point de vue statistique si l'intervalle de confiance de 95 % ne comprend pas la valeur 0.

3. Mesures complexes de l'inégalité

Une mesure complexe des inégalités tient compte des données de tous les sous-groupes de population (p. ex. inégalité à l'échelle de tous les groupes de revenus). Il s'agit d'une valeur unique qui rend compte du niveau d'inégalité.

La **réduction potentielle des taux (RPT)** est une mesure *relative* de la réduction potentielle du taux d'un indicateur de santé si chaque sous-groupe de population affichait le même taux que le sous-groupe ayant le taux le plus souhaitable. Elle est aussi communément appelée fraction étiologique du risque ou risque attribuable à la population¹.

Exemple : RPT = 15,0 %

Interprétation : Au cours d'une année donnée, 15 % des hospitalisations liées à l'affection X auraient pu être évitées si les Canadiens de tous les quintiles de revenu avaient affiché le même taux d'hospitalisations que ceux du quintile de revenu le plus élevé.

La RPT et son **intervalle de confiance de 95 %** sont calculés de la manière suivante, pour un facteur de stratification à 5 catégories (p. ex. les quintiles de revenu) où la catégorie 5 est le sous-groupe affichant le taux le plus souhaitable :

$$\text{RPT} = \frac{\sum_{i=1}^5 P_i \left(\frac{\text{taux}_i}{\text{taux}_5} - 1 \right)}{1 + \sum_{i=1}^5 P_i \left(\frac{\text{taux}_i}{\text{taux}_5} - 1 \right)} \times 100 \%$$

où P_i est la proportion de la population dans la catégorie i .

La limite de confiance inférieure (LCI) et la limite de confiance supérieure (LCS) sont donc calculées comme suit :

$$\text{LCI}_{\text{RPT}} = 1 - \frac{1}{P_5 + e^{(\log(\text{ratio des taux}_{1:4,5}) - 1,96\sqrt{\text{variance}(\text{ratio des taux}_{1:4,5})})}}$$

$$\text{LCS}_{\text{RPT}} = 1 - \frac{1}{P_5 + e^{(\log(\text{ratio des taux}_{1:4,5}) + 1,96\sqrt{\text{variance}(\text{ratio des taux}_{1:4,5})})}}$$

où Ratio des taux_{1:4, 5} est le ratio entre la somme de la proportion de la population (P_i), multipliée par le taux normalisé selon l'âge (Taux_{*i*}) de la première catégorie et le taux de la catégorie de référence, calculé comme suit :

$$\frac{\sum_{i=1}^4 P_i \text{taux}_i}{\text{taux}_5}$$

avec la variance

$$\frac{\sum_{i=1}^4 P_i^2 \times \text{var}(\text{taux}_i)}{(\sum_{i=1}^4 P_i \text{taux}_i)^2} + \frac{\text{variance}(\text{taux}_5)}{\text{taux}_5^2}$$

La RPT est considérée comme significativement différente de l'hypothèse nulle (RPT = 0) du point de vue statistique si l'intervalle de confiance de 95 % ne comprend pas la valeur 0.

Le **NPRP** est une mesure *absolue* de la réduction potentielle du nombre de cas associés à un indicateur de santé si chaque sous-groupe de population affichait le même taux que le sous-groupe ayant le taux le plus souhaitable. Cette mesure rend compte du gradient de l'inégalité à l'échelle de plusieurs catégories, comme les quintiles de revenu.

Exemple : NPRP = 7 300

Interprétation : Au cours d'une année donnée, 7 300 hospitalisations liées à l'affection X auraient pu être évitées si les Canadiens de tous les niveaux de revenu avaient affiché le même taux d'hospitalisations que ceux du niveau de revenu le plus élevé.

Le NPRP est lié à la RPT de la manière suivante :

$$\text{NPRP} = \text{taux global normalisé de l'indicateur} \times N_{\text{population totale}} \times \text{RPT}$$

Points à considérer lors de la déclaration des mesures complexes de l'inégalité

- La RPT s'applique idéalement lorsque des taux bas sont recherchés et que le ratio des taux est supérieur à 1. Si un taux élevé est souhaitable et que le ratio des taux est inférieur à 1, envisagez de calculer l'amélioration potentielle du taux (APT, aussi appelée fraction préventive).
- Envisagez de déclarer la RPT et le NPRP comme nuls si la RPT est négative et différente de 0 du point de vue statistique (c.-à-d. que les limites de confiance inférieure et supérieure sont négatives).
- Envisagez de déclarer le NPRP comme un nombre approximatif (p. ex. arrondis à la centaine près).

Annexe 1 : Macro SAS — %Calculate_Stratified_Rates

Nom de la macro

Calculate_Stratified_Rates

Objectif

Calculer les taux bruts et normalisés selon l'âge, stratifiés par quintile de revenu ou par emplacement géographique (région urbaine ou rurale/éloignée). Le revenu et l'emplacement géographique (urbain ou rural/éloigné) sont 2 facteurs de stratification de l'équité couramment utilisés pour mesurer les inégalités en santé. Les taux stratifiés sont aussi calculés en fonction du sexe (hommes, femmes et les 2 sexes) et du niveau de déclaration (à l'échelle du pays ou des provinces et territoires). Ce code peut être modifié pour calculer les taux stratifiés en fonction d'autres facteurs de stratification de l'équité.

Fichiers d'entrée

Cette macro requiert 3 fichiers d'entrée :

1. Cas liés à l'indicateur — par facteur de stratification de l'équité, niveau de déclaration, sexe et groupe d'âge;
2. Dénominateurs de population — par facteur de stratification de l'équité, niveau de déclaration, sexe et groupe d'âge;
3. Population de référence — nous utilisons la population canadienne de 2011 par tranches d'âge de 5 ans.

Les fichiers d'entrée doivent être formatés selon la convention indiquée dans le document Excel [Mesurer les inégalités en santé : trousse d'outils — formats des fichiers d'entrée des macros SAS](#).

Fichier de sortie

&indicator._&YR._&AGEGPL._&AGEGPU._&equity_stratifier._rates — Taux bruts et normalisés selon l'âge du facteur de stratification de l'équité choisi (revenu ou emplacement géographique) et le niveau de déclaration.

Paramètres (tous requis)

Avant de lancer la macro, vous devez définir les variables `&indata_directory` et `&outdata_directory` et régler les paramètres ci-dessous.

- *indicator* — Nom de l'indicateur de santé mesuré; il peut s'agir de n'importe quel nom de votre choix.
- *yr* — Année.
- *reporting_level* — Niveau de déclaration géographique (doit correspondre au code de la province).
- *equity_stratifier* — Variable pour laquelle vous désirez calculer les taux stratifiés. Vous avez actuellement 2 options : QAIPPE pour le quintile de revenu ou URR pour l'emplacement géographique (région urbaine ou rurale/éloignée).
- *infile* — Nom du fichier d'entrée des cas liés à l'indicateur.
- *stdpop* — Nom du fichier de population de référence.
- *popfile* — Nom du fichier de population correspondant au revenu ou à l'emplacement géographique (c.-à-d. les estimations démographiques par quintile de revenu ou par région urbaine ou rurale/éloignée).
- *popvar* — Chiffres concernant la population provenant du fichier de population (doit correspondre à la population de la province).
- *outfile* — Nom du fichier de sortie.
- *AGEGPL* — Limite d'âge inférieure de l'indicateur de santé; doit respecter le format 01, 02, [...], 14, 15 tel que précisé dans la variable de l'âge (p. ex. si la limite d'âge inférieure pour votre indicateur de santé est 0 an, cette valeur doit être 01).
- *AGEGPU* — Limite d'âge supérieure de l'indicateur de santé; doit respecter le format 01, 02, [...], 14, 15 tel que précisé dans la variable de l'âge (p. ex. si la limite d'âge supérieure pour votre indicateur de santé est 74 ans, cette valeur doit être 15).
- *MULTIPLIER* — Valeur pour laquelle vous voulez présenter les taux (p. ex. le MULTIPLIER est 100 000 pour les taux par 100 000).
- *ROUNDUNIT* — Valeur d'arrondissement (p. ex. pour arrondir à une décimale, précisez la ROUNDUNIT = .01).

Invocation de la macro

%MACRO Calculate_Stratified_Rates (indicator =, yr =, reporting_level =, equity_stratifier =, infile =, stdpop =, popfile =, popvar =, outfile =, AGEGPL =, AGEGPU =, MULTIPLIER =, ROUNDUNIT =);

Étapes

1. Préparer les cas liés à l'indicateur en créant des données agrégées par niveau de déclaration, groupe d'âge, sexe et facteur de stratification de l'équité.
2. Préparer le fichier de population de référence qui servira à la normalisation selon l'âge.
3. Préparer les dénominateurs de population de la tranche d'âge visée.
4. Lier les 3 fichiers de données ci-dessus à l'échelle des provinces et des territoires par groupe d'âge, sexe et facteur de stratification de l'équité.
5. Calculer les taux stratifiés bruts et suivre la procédure suivante pour calculer les taux stratifiés normalisés selon l'âge :
 - a) Calculer les taux propres à chaque groupe d'âge.
 - b) Multiplier les taux des groupes d'âge de la population étudiée par le nombre de personnes dans chaque groupe d'âge de la population de référence pour obtenir la valeur attendue de chaque groupe d'âge.
 - c) Additionner les valeurs attendues de tous les groupes d'âge. Enfin, pour obtenir les taux ajustés selon l'âge, diviser la valeur attendue totale par la population de référence.
 - d) Calculer la variance et utiliser cette valeur pour calculer les intervalles de confiance.
6. Calculer les taux globaux bruts et normalisés selon l'âge. Notez que les taux globaux correspondent à l'ensemble des catégories d'un facteur de stratification de l'équité (p. ex. pour le facteur de stratification du revenu, le taux global correspond aux quintiles 1 à 5). Ainsi, les cas qui ne sont pas associés à une catégorie de revenu (p. ex. à cause d'un code postal manquant) seront exclus du taux global.
7. Créer un tableau avec les données calculées.

Code de la macro

```

/*=====*/
/* - Define input library
/*=====*/
libname data "&indata_directory.";
libname out "&outdata_directory.";

/*=====*/
/* - Define %include statements
/*=====*/

%MACRO Calculate_Stratified_Rates(indicator= , yr= , reporting_level= ,
equity_stratifier= , infile=, stdpop=, popfile=, popvar=,
outfile=,AGEGPL=, AGEGPU=, MULTIPLIER=, ROUNDUNIT=);

/*****/
/* STEP 1: Prepare data set of individual-level indicator cases */
/*****/

/*****/
/* 1A. Create count variable for the indicator data set and output
required year(s) and 5-year age groups
/*****/
data &indicator.&yr._&AGEGPL._&AGEGPU.;
    set &infile.;
    * Create a count variable for the indicator;
    &indicator.=1;
    * Output required year and age groups only;
    if year eq &yr. and ("&AGEGPL."<=AGE_GROUP_CODE<="&AGEGPU.") then
output;
run;

/*****/
/* 1B. Create aggregated indicator data sets - national and by
province/territory */
/* ==> Prepare aggregated counts from indicator data set (infile) by age
group, sex (male/female/overall) and equity stratifier.
/* Do this for Canada and by reporting level (&reporting_level.)
/*****/
* Sum indicator cases by sex, age group and equity stratifier for all of
Canada;
proc sql;
    create table &indicator._canada as
    select sex, AGE_GROUP_CODE, &equity_stratifier., SUM(&indicator.) as
cases
    from &indicator.&yr._&AGEGPL._&AGEGPU.
    group by sex, AGE_GROUP_CODE, &equity_stratifier.
    OUTER UNION CORR
    select "3" AS sex, AGE_GROUP_CODE, &equity_stratifier.,
SUM(&indicator.) as cases
    from &indicator.&yr._&AGEGPL._&AGEGPU.
    group by AGE_GROUP_CODE, &equity_stratifier.
    ORDER BY SEX, AGE_GROUP_CODE, &equity_stratifier.;
QUIT;

```

```

/* Define Canada as '99' */
data &indicator._canada;
    set &indicator._canada;
    &reporting_level.="99";
run;

* Sum indicator cases by sex, age group and equity stratifier by reporting
level from &reporting_level.;
proc sql;
    create table &indicator._&reporting_level. as
    select &reporting_level., sex, AGE_GROUP_CODE, &equity_stratifier.,
SUM(&indicator.) as cases
    from &indicator.&yr._&AGEGPL._&AGEGPU.
    group by &reporting_level., sex, AGE_GROUP_CODE, &equity_stratifier.
    OUTER UNION CORR
    select &reporting_level., "3" AS sex, AGE_GROUP_CODE,
&equity_stratifier., SUM(&indicator.) as cases
    from &indicator.&yr._&AGEGPL._&AGEGPU.
    group by &reporting_level., AGE_GROUP_CODE, &equity_stratifier.
    ORDER BY &reporting_level., SEX, AGE_GROUP_CODE, &equity_stratifier.;
QUIT;

data &indicator._&reporting_level.;
    set &indicator._&reporting_level. &indicator._canada;
run;

/*****
/*          STEP 2. Prepare standard population file          */
*****/

/*****
/*          2A. Sum the standard population by indicator age groups within
age range of the indicator and keep the total population
/*          ==> Multiply each of the age-specific rates by the proportion of
the 2011 population belonging to the particular age group (called the
standard population weight)
*****/

proc sql;
    create table stdpop_agegroup as
    select AGE_GROUP_CODE, standard_pop_cnt as stdpop, sum(standard_pop_cnt) as
agegroup_stdpop
    from &stdpop. (where=("&AGEGPL."<=AGE_GROUP_CODE<="&AGEGPU."));
quit;
proc sort data=stdpop_agegroup; by AGE_GROUP_CODE; run;

proc sql;
    create table stdpop_population as
    select AGE_GROUP_CODE, sum(standard_pop_cnt) as total_stdpop
    from &stdpop. (where=("&AGEGPL."<=AGE_GROUP_CODE<="&AGEGPU."));
quit;
proc sort data=stdpop_population; by AGE_GROUP_CODE; run;

data stdpop_new;
    merge stdpop_agegroup (in=a) stdpop_population(in=b);
    by AGE_GROUP_CODE;

```

```

        if a=1 then do;
            weight= stdpop/agegroup_stdpop;
            output;
        end;
        drop total_stdpop;
run;

/*****
/*          STEP 3. Prepare population data          */
*****/

/*****
/*          3A. Prepare population denominators for the specific age range of
indicator */
*****/
data new_popfile;
    set &popfile.;
    if population_year eq &yr. and
("&AGEGPL."<=AGE_GROUP_CODE<="&AGEGPU.") then output;
run;

/*****
/*          STEP 4. Join data sets from steps 1 to 3
*/
*****/
/*****
/*          4A. Join standard population and population file
*****/
proc sql;
    create table new_popfile2 as
    select a.*, b.stdpop, b.weight
    from new_popfile as a
    right join stdpop_new as b
    on a.AGE_GROUP_CODE=b.AGE_GROUP_CODE
    order by &reporting_level., a.sex, b.AGE_GROUP_CODE, a.&equity_stratifier.
;
quit;

/*****
/*          4B. Add in aggregated numerator cases and roll up by equity
stratifier, reporting_level, sex and age
*****/
proc sql;
    create table count_pop_&equity_stratifier. as
    select a.cases as count, b.*
    from &indicator._&reporting_level. as a
    right join new_popfile2 as b
on a.AGE_GROUP_CODE=b.AGE_GROUP_CODE and a.sex=b.sex and
a.&equity_stratifier.= b.&equity_stratifier. and
a.&reporting_level.=b.&reporting_level.
    order by &reporting_level., &equity_stratifier., sex, AGE_GROUP_CODE;
quit;

```

```

/*****/
/*      4C. Create new population file rolling up all income quintiles or
geographic locations (urban and rural/remote) to be used in calculating
overall age-standardized rates
/*****/
proc sql;
    create table new_popfile3 as
    select population_year, province_code, sex, age_group_code,
sum(&popvar.) as &popvar., stdpop, weight
    from new_popfile2
    group by population_year, province_code, sex, age_group_code, stdpop,
weight;
quit;

/*****/
/*      4D. Add in aggregated numerator cases by reporting level, sex and
age only - will use this for calculating overall rates (e.g., rates for all
income quintiles combined or urban and rural/remote combined)
/*****/

proc sql;
    create table &indicator._&reporting_level._all
    as select &reporting_level., sex, AGE_GROUP_CODE, sum(cases) as cases
    from &indicator._&reporting_level.
    group by &reporting_level., sex, AGE_GROUP_CODE;
quit;

proc sql;
    create table count_pop as
    select a.cases as count, b.*
    from &indicator._&reporting_level._all as a
    right join new_popfile3 as b
    on a.AGE_GROUP_CODE=b.AGE_GROUP_CODE and a.sex=b.sex and
a.&reporting_level.=b.&reporting_level.
    order by &reporting_level., sex, AGE_GROUP_CODE;
quit;

/*****/
/*      STEP 5. Calculate crude and age-standardized rates stratified by
&equity_stratifier
/*****/

/*****/
/*      5A. Calculate crude, expected, variance and standardized numerator
count for each row in data set
/*      ==> Variance calculation is based on a binomial distribution
/*      ==> Standardized numerator count (std_num) is required in
Calculate_Inequality_Measures macro */
/*****/
%if &popvar. ne 0 %then %do;
    data &indicator._rates_&YR._&AGEGPL._&AGEGPU.;
    set count_pop_&equity_stratifier.;
    if count=. then count=0;
    crude=(count/&popvar.)*&MULTIPLIER.;
    expect=crude*weight;
    variance=((weight*weight)*crude*(&MULTIPLIER.-crude))/&popvar.;

```

```

        std_num=count/&popvar.*stdpop;
run;
%end;

%if &popvar. = 0 %then %do;
data &indicator._rates_&YR._&AGEGPL._&AGEGPU.;
    set count_pop_&equity_stratifier.;
    if count=. then count=0;
    crude=0;
    expect=crude*weight;
    variance=((weight*weight)*crude*(&MULTIPLIER.-crude))/&popvar.;
    std_num=count/&popvar.*stdpop;
run;
%end;

/*****
/* 5B. Calculate age-standardized rates (ASR) by equity stratifier
/* ==> Sum counts, total population, crude rate, standard
numerator, expected and variance by
/* ==> geographic reporting level, sex and equity stratifier
*****/
proc sql;
    create table &indicator._ASR_&YR._&AGEGPL._&AGEGPU._&equity_stratifier. as
        select &reporting_level., sex, &equity_stratifier.,
            sum(count) as count, sum(&popvar.) as totalpop,
            sum(count)/sum(&popvar.)*&MULTIPLIER. as crude, sum(std_num) as
std_numer,
            sum(expect) as sum_exp, sum(variance) as variance
        from &indicator._rates_&YR._&AGEGPL._&AGEGPU.
    group by &reporting_level., sex, &equity_stratifier.;
quit;

/*****
/* 5C. Calculate confidence intervals for age-standardized rates by equity
stratifier
*****/
data &indicator._ASR_&YR._&AGEGPL._&AGEGPU._&equity_stratifier.;
    set &indicator._ASR_&YR._&AGEGPL._&AGEGPU._&equity_stratifier.;
    lci=sum_exp-1.96*sqrt(variance);
    uci=sum_exp+1.96*sqrt(variance);

    if lci<0 then do;
        p=sum_exp/&MULTIPLIER.;
        if p=0 then p=1/(&MULTIPLIER.*&MULTIPLIER.);
        logitp=log(p/(1-p));
        var=1/(p*p*(1-p)*(1-p))*(variance/(&MULTIPLIER.*&MULTIPLIER.));
        A=logitp-1.96*sqrt(var);
        B=logitp+1.96*sqrt(var);
        lci=&MULTIPLIER./(1+exp(-A));
        uci=&MULTIPLIER./(1+exp(-B));
        drop p logitp var A B;
    end;
%if &ROUNDUNIT. = 1 %then
        format crude rstd rlci ruci 8.;
%else
format asr asr_lci asr_uci 8&ROUNDUNIT.;;
rcrude= round(crude,&ROUNDUNIT.);

```



```

asr = round(sum_exp,&ROUNDUNIT.);
asr_lci = round(lci,&ROUNDUNIT.);
asr_uci = round(uci,&ROUNDUNIT.);
run;

/*****
/*   STEP 6.   Overall rates                               */
*****/

/*****
/*   6A. Calculate crude, expected, variance and standardized numerator
count for each row in data set
/*           ==> Variance calculation is based on a binomial distribution
/*           ==> Standardized numerator count (std_num) is required in
Calculate_Inequality_Measures macro */
*****/
%if &popvar. ne 0 %then %do;
  data &indicator._rates_&YR._&AGEGPL._&AGEGPU._ALL;
    set count_pop;
    if count=. then count=0;
    crude=(count/&popvar.)*&MULTIPLIER.;
    expect=crude*weight;
    variance=((weight*weight)*crude*(&MULTIPLIER.-crude))/&popvar.;
    std_num=count/&popvar.*stdpop;
  run;
%end;

%if &popvar. = 0 %then %do;
  data &indicator._rates_&YR._&AGEGPL._&AGEGPU._ALL;
    set count_pop;
    if count=. then count=0;
    crude=0;
    expect=crude*weight;
    variance=((weight*weight)*crude*(&MULTIPLIER.-crude))/&popvar.;
    std_num=count/&popvar.*stdpop;
  run;
%end;

/*****
/*   6B. Calculate age-standardized rates (ASR) by equity stratifier
/*           ==> Sum counts, total population, crude rate, standard numerator,
expected and variance by
/*           ==> geographic reporting level, sex and equity stratifier
*****/
proc sql;
  create table &indicator._ASR_&YR._&AGEGPL._&AGEGPU. as
    select &reporting_level., sex,
    sum(count) as count, sum(&popvar.) as totalpop,
    sum(count)/sum(&popvar.)*&MULTIPLIER. as crude, sum(std_num) as
std_num,
    sum(expect) as sum_exp, sum(variance) as variance
  from &indicator._rates_&YR._&AGEGPL._&AGEGPU._ALL
  group by &reporting_level., sex;
quit;

```

```

/*****
/* 6C. Calculate confidence intervals for overall age-standardized rates
*****/
data &indicator._ASR_&YR._&AGEGPL._&AGEGPU.;
  set &indicator._ASR_&YR._&AGEGPL._&AGEGPU.;
  lci=sum_exp-1.96*sqrt(variance);
  uci=sum_exp+1.96*sqrt(variance);

  if lci<0 then do;
    p=sum_exp/&MULTIPLIER.;
    if p=0 then p=1/(&MULTIPLIER.*&MULTIPLIER.);
    logitp=log(p/(1-p));
    var=1/(p*p*(1-p)*(1-p))*(variance/(&MULTIPLIER.*&MULTIPLIER.));
    A=logitp-1.96*sqrt(var);
    B=logitp+1.96*sqrt(var);
    lci=&MULTIPLIER./(1+exp(-A));
    uci=&MULTIPLIER./(1+exp(-B));
    drop p logitp var A B;
  end;
  %if &ROUNDUNIT. = 1 %then
    format crude rstd rlci ruci 8.;
  %else
  format asr asr_lci asr_uci 8&ROUNDUNIT.;;
  rcrude= round(crude,&ROUNDUNIT.);
  asr = round(sum_exp,&ROUNDUNIT.);
  asr_lci = round(lci,&ROUNDUNIT.);
  asr_uci = round(uci,&ROUNDUNIT.);
run;

/*****
/* STEP 7. Output data set
  Note: This data set can be used as the input data set for the
Calculate_Inequality_Measures macro to calculate inequality summary measures
by equity stratifier.
*/
*****/
* ASR by equity stratifier;
data &indicator._ASR_&YR._&AGEGPL._&AGEGPU._&equity_stratifier.;
  retain fiscal_year &reporting_level. sex &equity_stratifier count
  totalpop rcrude variance asr asr_lci asr_uci std_numer;
  set &indicator._ASR_&YR._&AGEGPL._&AGEGPU._&equity_stratifier.;
  fiscal_year=&yr.;
  keep fiscal_year &reporting_level. sex &equity_stratifier count
totalpop rcrude variance asr asr_lci asr_uci std_numer;
run;

* ASR overall;
data &indicator._ASR_&YR._&AGEGPL._&AGEGPU.;
  retain fiscal_year &reporting_level. sex &equity_stratifier count
  totalpop rcrude variance asr asr_lci asr_uci std_numer;
  set &indicator._ASR_&YR._&AGEGPL._&AGEGPU.;
  fiscal_year=&yr.;
  &equity_stratifier ="0";
  keep fiscal_year &reporting_level. sex &equity_stratifier count
totalpop rcrude variance asr asr_lci asr_uci std_numer;
run;

```

```

* Combine ASR overall and by equity stratifier;
data &indicator._&YR._&AGEGPL._&AGEGPU._&equity_stratifier._rates;
    set &indicator._ASR_&YR._&AGEGPL._&AGEGPU._&equity_stratifier.
&indicator._ASR_&YR._&AGEGPL._&AGEGPU.;
run;

proc sort data=&indicator._&YR._&AGEGPL._&AGEGPU._&equity_stratifier._rates;
by &reporting_level. sex &equity_stratifier.; run;

    data out.&outfile.;
        set &indicator._&YR._&AGEGPL._&AGEGPU._&equity_stratifier._rates;
    run;

%MEND Calculate_Stratified_Rates;

```

Annexe 2 : Macro SAS — %Calculate_Inequality_Measures

Nom de la macro

Calculate_Inequality_Measures

Objectif

Calculer les mesures des inégalités, dont le ratio des taux, la différence des taux, la réduction potentielle des taux (RPT) et le nombre de personnes à risque au sein de la population (NPRP), pour des indicateurs de santé stratifiés par quintile de revenu ou par emplacement géographique (région urbaine ou rurale/éloignée). Le revenu et l'emplacement géographique (urbain ou rural/éloigné) sont 2 facteurs de stratification de l'équité couramment utilisés pour la mesure des inégalités en santé. Cette macro calcule les taux d'inégalités liés au revenu et à l'emplacement géographique normalisés selon l'âge, selon le sexe et le niveau de déclaration (à l'échelle du pays ou des provinces et territoires).

Format du fichier d'entrée

Le fichier d'entrée doit être formaté selon la convention indiquée dans le document Excel

[Mesurer les inégalités en santé : trousse d'outils — formats des fichiers d'entrée des macros SAS.](#)

Notez que les données générées après l'exécution réussie de la macro

%Calculate_Stratified_Rates seront correctement formatées.

Fichier de sortie

2 fichiers de sortie seront créés après l'exécution de ce programme macro :

rd_rr_&indicator.&yr_by_&equity_stratifier. — ratio des taux et différence des taux selon le facteur de stratification et le niveau de déclaration

prr_pin_&indicator.&yr_by_&equity_stratifier. — RPT et NPRP selon le facteur de stratification et le niveau de déclaration

Paramètres (tous requis)

- *data* — Nom du tableau de données d'entrée de la macro.
- *indicator* — Nom de l'indicateur de santé mesuré; il peut s'agir de n'importe quel nom de votre choix.
- *yr* — Année.
- *equity_stratifier* — Variable pour laquelle vous désirez calculer les mesures des inégalités. Vous avez actuellement 2 options : QAIPPE pour le quintile de revenu ou URR pour l'emplacement géographique (région urbaine ou rurale/éloignée).
- *reporting_level* — Niveau de déclaration géographique (doit correspondre au code de la province).
- *sex* — Variable de la répartition de l'indicateur selon le sexe; obligatoire même si le rapport s'intéresse aux 2 sexes (dans ce cas, entrez une valeur factice).
- *outlib* — Nom de la librairie de l'emplacement de sortie.
- *multiplier* — Valeur pour laquelle vous voulez présenter les taux (p. ex. le *multiplier* est 100 000 pour les taux par 100 000).

Invocation de la macro

%MACRO Calculate_Inequality_Measures (data =, indicator =, yr =, equity_stratifier =, reporting_level =, sex =, outlib =, multiplier =);

Étapes

1. Créer des sous-ensembles des données d'entrée qui serviront à comparer les sous-groupes présentant les taux les plus souhaitables et les moins souhaitables.
2. Calculer le ratio des taux et la différence des taux.
3. Calculer la RPT et le NPRP.
4. Créer des tableaux de données comprenant toutes les mesures sommaires.

Code de la macro

```

%MACRO Calculate_Inequality_Measures(data=, indicator=, yr=,
equity_stratifier=, reporting_level=, sex=, outlib=, multiplier=);

/*****
/*          STEP 1: Subset data          */
*****/

%if &equity_stratifier. = QAIPPE %then %do;
/*****
/** 1A. Subset data by income stratifier to keep Q1 and Q5 ***/
*****/

data &indicator.&yr._lo;
set &data.;
keep &reporting_level. &sex. QAIPPE asr variance asr_uci asr_lci;
if QAIPPE = 1;
run;

data &indicator.&yr._hi;
set &data.;
keep &reporting_level. &sex. QAIPPE asr variance asr_uci asr_lci;
if QAIPPE = 5;
run;
%end;

%if &equity_stratifier. = URR %then %do;
/*****
/** 1B. Subset data by geography stratifier to keep urban versus
rural_remote ***/
*****/

data &indicator.&yr._hi;
set &data.;
keep &reporting_level. &sex. asr variance asr_uci asr_lci;
if urr = 1;
run;

data &indicator.&yr._lo;
set &data.;
keep &reporting_level. &sex. asr variance asr_uci asr_lci;
if urr = 2;
run;
%end;

/*****
/*  STEP 2: Calculate RR and RD based on stratified data */
*merge back together into flat file to calculate RR, RD*;
*****/

/*****
*2A. Output table with RR + RD*;
*****/

```

```

proc sql;
create table &outlib..rd_rr_&indicator._&yr._by_&equity_stratifier. as
select &yr. as fiscal_year, a.&reporting_level., a.&sex., a.asr as
lo_std_rate, a.asr_lci as lo_lcl, a.asr_uci as lo_ucl, b.asr as hi_std_rate,
b.asr_lci as hi_lcl, b.asr_uci as hi_ucl, a.variance as lo_var, b.variance as
hi_var,
(a.asr/b.asr) as rr, ((lo_var/(lo_std_rate**2))+(hi_var/(hi_std_rate**2))) as
var_logdrr,
exp(log(calculated rr)-(1.96*sqrt(calculated var_logdrr))) as lcl_rr,
exp(log(calculated rr)+(1.96*sqrt(calculated var_logdrr))) as ucl_rr,
(a.asr - b.asr) as rd, (lo_var+hi_var) as var_rd,
(calculated rd-(1.96*sqrt(calculated var_rd))) as lcl_rd,
(calculated rd+(1.96*sqrt(calculated var_rd))) as ucl_rd
from &indicator.&yr._lo as a inner join &indicator.&yr._hi as b
on a.&reporting_level. = b.&reporting_level. and a.&sex. = b.&sex.;

/*****
/* STEP 3: Calculate PRR and PIN based on stratified data (income stratifier
only) */
*****/

/*****
*3A. Calculate PRR*;
*Use study pop as the population proportion*;
*****/

%if &equity_stratifier. = QAIPPE %then %do;
proc sql;
*create column of quintile 5 std rate and proportions*;
create table prr_&indicator.&yr._a as
select &reporting_level., &sex., QAIPPE, totalpop, asr, variance,
(totalpop/sum(totalpop)) as pop_portion
from &data.
where QAIPPE ne '0'
group by &reporting_level., &sex.;

create table prr_&indicator.&yr._b as
select a.&reporting_level., a.&sex., a.QAIPPE, a.totalpop, a.asr, a.variance,
a.pop_portion,
b.asr as tot_std_hi, a.asr/tot_std_hi as rate_ratio_hi
from prr_&indicator.&yr._a as a inner join &indicator.&yr._hi as b
on a.&reporting_level. = b.&reporting_level. and a.&sex. = b.&sex.;

/*****
*3B. Calculate PRR*;
*****/

create table prr_&indicator.&yr. as
select &reporting_level., &sex., sum(pop_portion*(rate_ratio_hi-
1))/(1+sum(pop_portion*(rate_ratio_hi-1))) as prr
from prr_&indicator.&yr._b
group &reporting_level., &sex.;

*Calculate CIs for PRR*;
create table rate_low_&indicator.&yr. as
select &reporting_level., &sex., sum((pop_portion**2)*variance) as
variance_low, sum(pop_portion*asr) as sum_low

```

```

from prr_&indicator.&yr._b
where QAIPE ne '5'
group &reporting_level., &sex.;

create table rate_hi_&indicator.&yr. as
select &reporting_level., &sex., variance as variance_hi, asr as rate_hi,
pop_portion as pop_portion_hi
from prr_&indicator.&yr._b
where QAIPE = '5';

create table prr_ci_&indicator.&yr. as
select a.&reporting_level., a.&sex., prr,
(sum_low/rate_hi) as rate_low,
((variance_low/sum_low**2)+(variance_hi/rate_hi**2)) as varlog_low,
exp(log(calculated_rate_low)-1.96*sqrt(calculated_varlog_low)) as lcl_low,
exp(log(calculated_rate_low)+1.96*sqrt(calculated_varlog_low)) as ucl_low,
1-(1/(pop_portion_hi+calculated_lcl_low)) as lcl_prr,
1-(1/(pop_portion_hi+calculated_ucl_low)) as ucl_prr
from prr_&indicator.&yr. as a, rate_low_&indicator.&yr. as b,
rate_hi_&indicator.&yr. as c
where a.&reporting_level.=b.&reporting_level.=c.&reporting_level. and
a.&sex.=b.&sex.=c.&sex.;

/*****
/*STEP 3: Calculate PRR and PIN based on stratified data (income stratifier
only) */
*****/

create table prr_pin_&indicator.&yr._by_&equity_stratifier. as
select &yr. as fiscal_year, a.&reporting_level., a.&sex.,
%if &equity_stratifier. = QAIPE %then %do;
    qaipe,
%end;
%if &equity_stratifier. = URR %then %do;
    urr,
%end;
prr, lcl_prr, ucl_prr, std_numer*prr as pin
from &data. as a left join prr_ci_&indicator.&yr. as b
on a.&reporting_level.=b.&reporting_level. and a.&sex.=b.&sex.;

proc sort data=prr_pin_&indicator.&yr._by_&equity_stratifier.
out=&outlib..prr_pin_&indicator.&yr._by_&equity_stratifier. nodup;
by &reporting_level. &sex.;
run;
%end;

/*****
/* STEP 4: Output final data as a table with all summary measures */
*****/

%if &equity_stratifier. = QAIPE %then %do;
proc sql;
create table &indicator._inequality_summary as
select a.fiscal_year, b.&reporting_level. as jurisdiction, c.qaipe, a.&sex.,
b.count, b.rcrude as crude_rate, b.asr,
b.asr_lci, b.asr_uci , a.rr, a.lcl_rr, a.ucl_rr, a.rd, a.lcl_rd, a.ucl_rd,
c.prr, c.lcl_prr, c.ucl_prr, c.pin

```



```

from &outlib..rd_rr_&indicator._&yr._by_&equity_stratifier. as a, &data. as
b, &outlib..prrr_pin_&indicator._&yr._by_&equity_stratifier. as c
where a.&reporting_level.=b.&reporting_level.=c.&reporting_level. and
a.&sex.=b.&sex.=c.&sex. and b.qaippe = c.qaippe
group by c.qaippe;
%end;

%if &equity_stratifier. = URR %then %do;
proc sql;
create table &indicator._inequality_summary as
select a.fiscal_year, b.&reporting_level. as jurisdiction, b.urr, a.&sex.,
b.count, b.rcrude as crude_rate, b.asr,
b.asr_lci, b.asr_uci , a.rr, a.lcl_rr, a.ucl_rr, a.rd, a.lcl_rd, a.ucl_rd
from &outlib..rd_rr_&indicator._&yr._by_&equity_stratifier. as a, &data. as b
where a.&reporting_level.=b.&reporting_level. and a.&sex.=b.&sex.
group by b.urr;
%end;

%let stratifier2 = %substr(&equity_stratifier.,1,3);

data &outlib..&indicator._&stratifier2._inequality_summary;
retain fiscal_year jurisdiction Stratifier sex count crude_rate asr
asr_lci asr_uci rr lcl_rr ucl_rr rd lcl_rd ucl_rd
%if &equity_stratifier. = QAIPPE %then %do;
prrr lcl_prr ucl_prr pin
%end;
;
length stratifier $12.;
set &indicator._inequality_summary;
%if &equity_stratifier. = QAIPPE %then %do;
if qaippe = '0' then Stratifier = '0 (overall)';
if qaippe = '1' then Stratifier = '1 (lowest)';
if qaippe = '2' then Stratifier = '2';
if qaippe = '3' then Stratifier = '3';
if qaippe = '4' then Stratifier = '4';
if qaippe = '5' then Stratifier = '5 (highest)';
label Stratifier = 'Income Quintile';
%end;
%if &equity_stratifier. = URR %then %do;
if urr eq 0 then Stratifier = '0 (overall)';
if urr eq 1 then Stratifier = 'Urban';
if urr eq 2 then Stratifier = 'Rural/Remote';
%end;
label fiscal_year = 'Fiscal Year'
jurisdiction = 'Jurisdiction'
sex = 'Sex'
count = 'Number of Indicator Cases'
crude_rate = "Crude Rate per &MULTIPLIER. population"
asr = "Age-standardized rate per &MULTIPLIER. population"
asr_lci = 'Lower confidence limit of the age-standardized rate'
asr_uci = 'Upper confidence limit of the age-standardized rate'
rr = 'Rate Ratio'
lcl_rr = 'Lower confidence limit of the rate ratio'
ucl_rr = 'Upper confidence limit of the rate ratio'
rd = "Rate difference per &MULTIPLIER. population"
lcl_rd = 'Lower confidence limit of the rate difference'
ucl_rd = 'Upper confidence limit of the rate difference'

```

```
%if &equity_stratifier. = QAIPPE %then %do;
    prr      = 'Potential Reduction Rate'
    lcl_prr  = 'Lower confidence limit of the Potential Rate
Reduction'
    ucl_prr  = 'Upper confidence limit of the Potential Rate
Reduction'
    pin      = "Population Impact Number per &MULTIPLIER. population";
%end;
keep fiscal_year jurisdiction Stratifier sex count crude_rate asr
asr_lci asr_uci rr lcl_rr ucl_rr rd lcl_rd ucl_rd
%if &equity_stratifier. = QAIPPE %then %do;
prr lcl_prr ucl_prr pin
%end;
;
%if &equity_stratifier. = URR %then %do;
drop urr;
%end;
run;

proc sort data=&outlib..&indicator._&stratifier2._inequality_summary;
    by fiscal_year jurisdiction sex Stratifier;
run;
%mend;
```

Référence

1. Rockhill B, Newman B, Weinberg C. [Use and misuse of population attributable fractions.](#) *American Journal of Public Health.* January 1998.



ICIS Ottawa

495, chemin Richmond
Bureau 600
Ottawa (Ont.)
K2A 4H6
613-241-7860

ICIS Toronto

4110, rue Yonge
Bureau 300
Toronto (Ont.)
M2P 2B7
416-481-2002

ICIS Victoria

880, rue Douglas
Bureau 600
Victoria (C.-B.)
V8W 2B7
250-220-4100

ICIS Montréal

1010, rue Sherbrooke Ouest
Bureau 602
Montréal (Qc)
H3A 2R7
514-842-2226

icis.ca

18433-0918

